
How to Produce a TABULATE Procedure Table with All Zeros 

ABSTRACT 
The TABULATE procedure in SAS® is frequently used to create appealing reports.  However, there is 
currently no easy way in PROC TABULATE to produce a table showing all zeroes when the data set does 
not have any cases satisfying the WHERE statement.  This paper demonstrates a technique that works 
around this limitation and enables you to produce the desired output whether or not any cases satisfy your 
filtering criteria. 
 
INTRODUCTION 
You have just been handed an assignment in which you need to create a PROC TABULATE report based 
on data collected from a clinical study.  The report is based on gender, ethnicity and racial categories.  Only 
a subset of the cases are used for the report, and you know that there may not be any cases in this subset 
at the beginning of the study.  If this happens, you would like to produce a table "shell" that would display 
zeroes for all of the cells.  
 
You start programming and executing code for this study using PROC TABULATE with the PRINTMISS 
option in SAS 8.2.  The report generated displays your data just as you had hoped:  when there are no 
cases in the subset, SAS produces a table containing all zeroes.  
 
Now, your company upgrades to SAS 9.1.3.  You execute your code and suddenly the table that was 
displaying all zeroes has vanished. What happened? Where did your table go? 
 
This paper will explain why the table disappeared, and demonstrate a workaround to produce the desired 
table in SAS 9.1.3. 
 
DATA EXAMPLES 
Below is the code to produce the data set used in this example.  
 

PROC FORMAT; 
  value ethnicf 

1 = 'Hispanic or Latino' 
2 = 'Not Hispanic or Latino' 
3 = 'Unknown (Individuals Not Reporting Ethnicity)'; 

 
  value racef 

1 = 'American Indian or Alaska Native' 
2 = 'Asian' 
3 = 'Native Hawaiian or Other Pacific Islander' 
4 = 'Black or African American' 
5 = 'White' 
6 = 'More Than One Race' 
7 = 'Unknown or Not Reported'; 

 
  value gndrf 

1 = 'Male' 
2 = 'Female' 
3 = 'Unknown or Not Reported'; 

RUN; 

DATA shelldata; 
  format ethlbl ethnicf. racelbl racef. gender gndrf.; 
  do ethcat = 1 to 2; 
    do ethlbl = 1 to 3; 
      do racelbl = 1 to 7; 
  do gender = 1 to 3; 
    output; 
  end; 

      end; 
    end; 
  end; 
RUN; 
 
DATA test; 
 input pt $ 1-3 ethlbl  gender  racelbl ; 
cards; 
x1  2 1 5 
x2  2 1 5 
x3  2 1 5 
x4  2 1 5 
x5  2 1 5 
x6  2 2 2  
x7  2 2 2 
x8  2 2 5 
x9  2 2 4 
x10 2 2 4 

RUN; 
 
DATA enroll; 
  set test; 
  if ethlbl = 1 then ethcat = 1; 
     else ethcat = 2; 
  format ethlbl ethnicf. racelbl racef. gender gndrf.; 
  label ethlbl = 'Ethnic Category' 
       racelbl = 'Racial Categories' 
        gender = 'Sex/Gender'; 

RUN; 
 
PROC TABULATE USING SAS 8.2  
Here is a simplified version of the program that was originally created using SAS 8.2 to produce the PROC 
TABULATE report. 
 
Since we are only interested in a subset of the cases, we want to specify a WHERE statement within the 
TABULATE procedure.  Using the PRINTMISS and MISSTEXT table options allowed the report to display 
rows filled with zeroes reflecting the missing data and the text from the MISSTEXT option. 
 

PROC TABULATE data=enroll classdata=shelldata f=8.0 missing ; 
 where ethcat = 1; 
 class  racelbl gender; 
 classlev racelbl gender; 
 keyword n pctn all; 
tables (racelbl all='Racial Categories:  Total of Hispanic or Latinos'), 
gender='Sex/Gender'*N=' ' all='Total'*n='' / printmiss misstext='0' 
box=[LABEL='']; 

 title1 font=arial color=darkblue h=1.5 'Inclusion Enrollment Report'; 
 title2 ' ';  
title3 font=arial color=darkblue h=1' PART B. HISPANIC ENROLLMENT REPORT:  
Number of Hispanic or Latinos Enrolled to Date (Cumulative)'; 

RUN; 

Here is the report that is generated when the above code is executed in SAS 8.2 and no persons 
characterizing themselves as Hispanic or Latino have yet enrolled in the study. 
 

 
 
As you can see the table is completely filled with zeroes — exactly what we want.   
   
PROC TABULATE USING SAS 9.1.3 
Here is the report that is generated when the above code is executed using SAS 9.1.3. 
 

 
 
As you can see, the page has only titles, no table. 

So, what is behind the mystery of the disappearing table?  We reported the problem to SAS Technical 
Support, and they responded that the behavior in 8.2 is a bug and the behavior in 9.1.3 is correct.  The 9.1.3 
behavior, in fact, goes back to SAS 6.  The 1990 TABULATE book, "SAS Guide to Tabulate Processing, 
Second Edition" says on page 48: 
 

Note that if an entire page of the report contains missing values, the PRINTMISS option has no 
effect on it; that page is not printed. 1 

 
Of course, what we really wanted was to print that page with all zeroes, rather than suppressing it.  
Fortunately, all is not lost.  In order to produce the table shell with all zeroes, macro processing can be used 
to modify the execution of PROC TABULATE when there are no observations that satisfy (or meet) the 
WHERE clause criteria.  This will create the desired “bugged“ table in SAS 9.1.3.   
 
PROC TABULATE MODIFICATION USING 9.1.3  
Here is how we modified the program to produce the desired output using SAS 9.1.3.  
 

%MACRO TAB_WHERE;  /* Include PROC TABULATE steps in a macro  */ 
 
Our first goal is to create a macro variable whose value will contain the number of observations meeting the 
WHERE clause criteria. 
  

  /*  PROC SQL step creates a macro variable whose  */ 
  /*   value will be the number of observations     */ 
  /*   meeting WHERE clause criteria.               */ 
  PROC SQL noprint; 
     select count(*) 
     into :numobs 
     from enroll 
     where ethcat=1; 
     QUIT; 

 
 
Once we obtain the number of observations meeting the WHERE criteria, we create a format to display all 
numeric values as zeroes.  If there are no observations that meet the criteria, we use the newly created 
format and omit the WHERE statement in the TABULATE procedure.  (We know that there is at least one 
case in the full data set.)  On the other hand, if there are observations that meet the criteria, we specify the 
WHERE statement and supply a general format for the data. 
 

/*  PROC FORMAT step to display all numeric values as zero.  */ 
 
  PROC FORMAT; 
     value allzero  low-high='       0'; 
  RUN; 
 
/*  Conditionally execute steps when no observations met criteria.    */ 
 
  %if &numobs=0 %then  
    %do; 
      %let fmt = allzero.;  /* Print all cell values as zeroes  */ 
      %let str = ;         /*No Cases in Subset - WHERE cannot be used */ 
    %end;  
  %else  
    %do; 
      %let fmt = 8.0; 
      %let str = where ethcat = 1;  
    %end; 

Coders’ CornerSUGI 31

 



 

 5

PROC TABULATE data=enroll classdata=shelldata missing format=&fmt; 
&str; 
format  racelbl racef. gender gndrf.; 
class  racelbl gender; 
classlev racelbl gender; 
keyword n pctn all; 
tables (racelbl all='Racial Categories:  Total of Hispanic or  Latinos'), 
gender='Sex/Gender'*N=' ' all='Total'*n='' /  printmiss  misstext='0' 
box=[LABEL=' '];  
title1 font=arial color=darkblue h=1.5 'Inclusion Enrollment Report'; 
title2 ' ';  
title3 font=arial color=darkblue h=1' PART B. HISPANIC ENROLLMENT REPORT:  
Number of Hispanic or Latinos Enrolled to Date (Cumulative)'; 

RUN; 
 
%MEND TAB_WHERE; 
 
%TAB_WHERE 

 
Here is the report that is generated when the above code is executed. 
 

 
 
 
Success!  As you can now see, this report looks identical to the one displayed that used SAS 8.2.  Please 
note that the PRINTMISS option has no effect on the report. 
 
CONCLUSION 
As is so often the case, you can produce almost any kind of report in SAS, even when there is no direct way 
to do so and you need to "trick" the system.  The example above illustrates one way to produce a PROC 
TABULATE table in SAS 9.1.3, even when all your data are missing.  In SAS, you can actually get 
something from nothing! 

